



**BC Cancer Agency**

CARE + RESEARCH

*An agency of the Provincial Health Services Authority*

**Cancer Surveillance & Outcomes**

# Basics of Proc Tabulate

Anky Lai

Cancer Surveillance & Outcomes (CSO)

Population Oncology

BC Cancer Agency

# Introduction

---

- What is Proc Tabulate?
  - A procedure that displays descriptive statistics in tabular format
- But aren't there other procs that do the same thing...




“I want summary stats on students’ ages and heights, overall and by gender.”

```
proc means data=sashelp.class;  
var age height;  
run;  
proc means data=sashelp.class;  
class sex;  
var age height; run;
```

Variable	N	Mean	Std Dev	Minimum	Maximum
Age	19	13.3157895	1.4926722	11.0000000	16.0000000
Height	19	62.3368421	5.1270752	51.3000000	72.0000000

Sex	N Obs	Variable	N	Mean	Std Dev	Minimum	Maximum
F	9	Age	9	13.2222222	1.3944334	11.0000000	15.0000000
		Height	9	60.5888889	5.0183275	51.3000000	66.5000000
M	10	Age	10	13.4000000	1.6465452	11.0000000	16.0000000
		Height	10	63.9100000	4.9379370	57.3000000	72.0000000



```
proc tabulate data=sashelp.class;  
class sex;  
var age height;  
table (age height)*(N MEAN STD MIN  
MAX), SEX ALL; run;
```

		Sex		All
		F	M	
Age	N	9	10	19
	Mean	13.22	13.40	13.32
	Std	1.39	1.65	1.49
	Min	11.00	11.00	11.00
	Max	15.00	16.00	16.00
Height	N	9	10	19
	Mean	60.59	63.91	62.34
	Std	5.02	4.94	5.13
	Min	51.30	57.30	51.30
	Max	66.50	72.00	72.00

# Introduction

---

- Why should you use it?
  - Payoff is in the output
  - Saves a lot of time
  - Reduces errors
  - Easy to format tables



**BC Cancer Agency**

CARE + RESEARCH

*An agency of the Provincial Health Services Authority*

**Cancer Surveillance & Outcomes**

# Outline

---

- PROC TABULATE basics
- Customize tables
- Export tables



**BC Cancer Agency**

CARE + RESEARCH

*An agency of the Provincial Health Services Authority*

**Cancer Surveillance & Outcomes**

# Proc Tabulate Basics

---

- Basic syntax:

(1) `proc tabulate data=dataset <options>;`

(2) `class variable(s) </option(s)>;` → variables used in PROC FREQ

**\*And/Or\***

`var variable(s) </option(s)>;` → variables used in PROC MEANS

(3) `table <page>,<row>,<column> </table-option(s)>; run;`



**BC Cancer Agency**

CARE + RESEARCH

*An agency of the Provincial Health Services Authority*

**Cancer Surveillance & Outcomes**

# Table Statement

---

- All variables listed in the TABLE statement must be listed in CLASS or VAR
- More than one TABLE statement can be used
- The order of the variables in the statement will be the same as the output table



# Table Statement

---

- Create 1, 2, or 3 dimensional tables:
  - 1 dimensional (no commas):  
Syntax: `table <column dimension> / <options>;`
  - 2 dimensional (1 comma):  
Syntax: `table <row dimension>,<column dimension>/<options>;`
  - 3 dimensional (2 commas):  
Syntax: `table <page>,<row>,<column>/<options>;`





# Table Statement

---

- Operators to use in each dimension:

- 1) Asterisk (\*)

- adding a classification variable

- adding another statistic

- 2) Parentheses ( )

- group elements and simplify coding and output

- 3) 'space'

- places output for each element immediately after the output of the preceding.

- 4) All

- calculate row or column totals



# Dataset

---

- Dataset: sashelp.CARS (N=428)
- Set up dataset as you would have it for PROC FREQ or PROC MEANS
- Examples done in SAS V9.3



# 1-Dimensional Tables

## ➤ Basic example

```
proc tabulate data=sashelp.CARS;  
  var MSRP;  
  class TYPE DRIVETRAIN;  
  table MSRP;  
  table TYPE DRIVETRAIN; run;
```

Output:

MSRP
Sum
14027638.00

Type						DriveTrain		
Hybrid	SUV	Sedan	Sports	Truck	Wagon	All	Front	Rear
N	N	N	N	N	N	N	N	N
3	60	262	49	24	30	92	226	110



BC Cancer Agency

CARE + RESEARCH

An agency of the Provincial Health Services Authority

Cancer Surveillance & Outcomes

# 1-Dimensional Tables

## ➤ Adding a classification variable

- Syntax: <variable name> \*<class variable>

```
proc tabulate data=sashelp.CARS ;  
  var MSRP;  
  class TYPE DRIVETRAIN;  
  table MSRP*TYPE MSRP*DRIVETRAIN; run;
```

Output:

MSRP						MSRP		
Sum						Sum		
Type						DriveTrain		
Hybrid	SUV	Sedan	Sports	Truck	Wagon	All	Front	Rear
59760.00	2087415.00	7800688.00	2615966.00	598593.00	865216.00	3356481.00	5600858.00	5070299.00



BC Cancer Agency

CARE + RESEARCH

An agency of the Provincial Health Services Authority

Cancer Surveillance & Outcomes

# 1-Dimensional Tables

## ➤ Adding another statistic

- Syntax: <variable name>**\*statistic-keyword**

Descriptive Statistics	Quantile Statistics
COLPCTN	MEDIAN   P50
PCTSUM	P1
COLPCTSUM	Q3   P75
MAX	P90
ROWPCTN	P95
MEAN	P5
ROWPCTSUM	P10
MIN	P99
STDDEV / STD	Q1   P25
N	QRANGE
STDERR	
NMISS	
SUM	
PAGEPCTSUM	
PCTN	
VAR	



# 1-Dimensional Tables

1) Add similar code to Table statement

```
proc tabulate data=sashelp.CARS;
class TYPE;
var MSRP ;
table MSRP*TYPE*sum MSRP*TYPE*mean;
run;
```

Output:

MSRP						MSRP					
Type						Type					
Hybrid	SUV	Sedan	Sports	Truck	Wagon	Hybrid	SUV	Sedan	Sports	Truck	Wagon
Sum	Sum	Sum	Sum	Sum	Sum	Mean	Mean	Mean	Mean	Mean	Mean
59760.00	2087415.00	7800688.00	2615966.00	598593.00	865216.00	19920.00	34790.25	29773.62	53387.06	24941.38	28840.53

2) Use parentheses

```
proc tabulate data=sashelp.cars;
class TYPE;
var MSRP ;
table MSRP*TYPE*(sum mean);
table MSRP*(sum mean)*TYPE ;run;
```

Output:

MSRP											
Type											
Hybrid		SUV		Sedan		Sports		Truck		Wagon	
Sum	Mean	Sum	Mean	Sum	Mean	Sum	Mean	Sum	Mean	Sum	Mean
59760.00	19920.00	2087415.00	34790.25	7800688.00	29773.62	2615966.00	53387.06	598593.00	24941.38	865216.00	28840.53

The SAS System

MSRP											
Sum						Mean					
Type						Type					
Hybrid	SUV	Sedan	Sports	Truck	Wagon	Hybrid	SUV	Sedan	Sports	Truck	Wagon
59760.00	2087415.00	7800688.00	2615966.00	598593.00	865216.00	19920.00	34790.25	29773.62	53387.06	24941.38	28840.53



BC Cancer Agency

CARE + RESEARCH

An agency of the Provincial Health Services Authority

Cancer Surveillance & Outcomes

# 1-Dimensional Tables

- Use 'ALL' to calculate overall statistics
  - Treat as a classification variable

```
proc tabulate data=sashelp.cars ;  
class TYPE;  
var MSRP ;  
table TYPE ALL (TYPE ALL)*MSRP*MEAN; run;
```

Output:

Type						All	Type						All
Hybrid	SUV	Sedan	Sports	Truck	Wagon		Hybrid	SUV	Sedan	Sports	Truck	Wagon	
							MSRP	MSRP	MSRP	MSRP	MSRP	MSRP	MSRP
N	N	N	N	N	N	N	Mean	Mean	Mean	Mean	Mean	Mean	Mean
3	60	262	49	24	30	428	19920.00	34790.25	29773.62	53387.06	24941.38	28840.53	32774.86



BC Cancer Agency

CARE + RESEARCH

An agency of the Provincial Health Services Authority

Cancer Surveillance & Outcomes

# Customizing 2-D Table

```
proc tabulate data=sashelp.cars ;
class DRIVETRAIN ORIGIN TYPE ;
var MSRP HORSEPOWER;
table ALL (DRIVETRAIN TYPE) * (N COLPCTN)
      (MSRP TYPE*HORSEPOWER) * (MEAN STD) ,
      (ORIGIN ALL) ;
run;
```

			Origin			All
			Asia	Europe	USA	
All	N		158	123	147	428
DriveTrain						
All	N		34	36	22	92
	ColPctN		21.52	29.27	14.97	21.50
Front	N		99	37	90	226
	ColPctN		62.66	30.08	61.22	52.80
Rear	N		25	50	35	110
	ColPctN		15.82	40.65	23.81	25.70
Type						
Hybrid	N		3	.	.	3
	ColPctN		1.90	.	.	0.70
SUV	N		25	10	25	60
	ColPctN		15.82	8.13	17.01	14.02
Sedan	N		94	78	90	262
	ColPctN		59.49	63.41	61.22	61.21
Sports	N		17	23	9	49
	ColPctN		10.76	18.70	6.12	11.45
Truck	N		8	.	16	24
	ColPctN		5.06	.	10.88	5.61
Wagon	N		11	12	7	30
	ColPctN		6.96	9.76	4.76	7.01
MSRP	Mean		24741.32	48349.80	28377.44	32774.86
	Std		11321.07	25318.60	11711.98	19431.72
Type						
Hybrid	Horsepower	Mean	92.00	.	.	92.00
		Std	18.52	.	.	18.52
SUV	Horsepower	Mean	214.16	263.10	246.56	235.82
		Std	48.70	52.66	58.68	56.23
Sedan	Horsepower	Mean	181.98	236.53	191.99	201.66
		Std	57.29	71.35	46.50	62.80
Sports	Horsepower	Mean	225.35	316.74	312.00	284.16
		Std	57.60	96.21	91.09	92.79
Truck	Horsepower	Mean	190.25	.	242.13	224.83
		Std	51.76	.	60.48	61.85
Wagon	Horsepower	Mean	185.64	218.17	165.71	194.00
		Std	69.47	63.71	44.20	63.79



BC Cancer Agency

CARE + RESEARCH

An agency of the Provincial Health Services Authority

Cancer Surveillance & Outcomes



# Customizing 2-D Table

- Print missing data

```
proc tabulate data=cars missing;
class DRIVETRAIN ORIGIN TYPE ;
var MSRP HORSEPOWER;
table ALL (DRIVETRAIN TYPE)*(N COLPCTN)
      (MSRP TYPE*HORSEPOWER)*(MEAN STD),
      (ORIGIN ALL);
run;
```

		Origin				All
			Asia	Europe	USA	
All	N	2	158	123	147	430
DriveTrain						
	N	2	.	.	.	2
	ColPctN	100.00	.	.	.	0.47
All	N	.	34	36	22	92
	ColPctN	.	21.52	29.27	14.97	21.40
Front	N	.	99	37	90	226
	ColPctN	.	62.66	30.08	61.22	52.56
Rear	N	.	25	50	35	110
	ColPctN	.	15.82	40.65	23.81	25.58
Type						
	N	2	.	.	.	2
	ColPctN	100.00	.	.	.	0.47
Hybrid	N	.	3	.	.	3
	ColPctN	.	1.90	.	.	0.70
SUV	N	.	25	10	25	60
	ColPctN	.	15.82	8.13	17.01	13.95
Sedan	N	.	94	78	90	262
	ColPctN	.	59.49	63.41	61.22	60.93
Sports	N	.	17	23	9	49
	ColPctN	.	10.76	18.70	6.12	11.40
Truck	N	.	8	.	16	24
	ColPctN	.	5.06	.	10.88	5.58
Wagon	N	.	11	12	7	30
	ColPctN	.	6.96	9.76	4.76	6.98
MSRP	Mean	.	24741.32	48349.80	28377.44	32774.86
	Std	.	11321.07	25318.60	11711.98	19431.72
Type						
	Horsepower	Mean	.	.	.	.
		Std	.	.	.	.
Hybrid	Horsepower	Mean	.	92.00	.	92.00
		Std	.	18.52	.	18.52
SUV	Horsepower	Mean	.	214.16	263.10	235.82
		Std	.	48.70	52.66	56.23
Sedan	Horsepower	Mean	.	181.98	236.53	201.66
		Std	.	57.29	71.35	62.80
Sports	Horsepower	Mean	.	225.35	316.74	284.16
		Std	.	57.60	96.21	92.79
Truck	Horsepower	Mean	.	190.25	.	224.83
		Std	.	51.76	.	61.85
Wagon	Horsepower	Mean	.	185.64	218.17	194.00
		Std	.	69.47	63.71	63.79



BC Cancer Agency

CARE + RESEARCH

An agency of the Provincial Health Services Authority

Cancer Surveillance & Outcomes

# Customizing 2-D Table

- Right align levels and statistic labels

```
proc tabulate data=sashelp.cars;
class DRIVETRAIN ORIGIN TYPE ;
classlev DRIVETRAIN TYPE/s=[just=right];
var MSRP HORSEPOWER;
table
ALL
(DRIVETRAIN TYPE) * (N = {s=[just=right]})
COLPCTN = {s=[just=right]})
(MSRP
TYPE*HORSEPOWER) * (MEAN= {s=[just=right]})
STD= {s=[just=right]}) , (ORIGIN ALL);
run;
```

		Origin			Total
		Asia	Europe	USA	
All	N	158	123	147	428
DriveTrain					
All	N	34	36	22	92
	ColPctN	21.52	29.27	14.97	21.50
Front	N	99	37	90	226
	ColPctN	62.66	30.08	61.22	52.80
Rear	N	25	50	35	110
	ColPctN	15.82	40.65	23.81	25.70
Type					
Hybrid	N	3	.	.	3
	ColPctN	1.90	.	.	0.70
SUV	N	25	10	25	60
	ColPctN	15.82	8.13	17.01	14.02
Sedan	N	94	78	90	262
	ColPctN	59.49	63.41	61.22	61.21
Sports	N	17	23	9	49
	ColPctN	10.76	18.70	6.12	11.45
Truck	N	8	.	16	24
	ColPctN	5.06	.	10.88	5.61
Wagon	N	11	12	7	30
	ColPctN	6.96	9.76	4.76	7.01
MSRP	Mean	24741.32	48349.80	28377.44	32774.86
	Std	11321.07	25318.60	11711.98	19431.72
Type					
Hybrid	Horsepower	Mean	92.00	.	92.00
		Std	18.52	.	18.52
SUV	Horsepower	Mean	214.16	263.10	235.82
		Std	48.70	52.66	56.23
Sedan	Horsepower	Mean	181.98	236.53	201.66
		Std	57.29	71.35	62.80
Sports	Horsepower	Mean	225.35	316.74	284.16
		Std	57.60	96.21	92.79
Truck	Horsepower	Mean	190.25	.	224.83
		Std	51.76	.	61.85
Wagon	Horsepower	Mean	185.64	218.17	194.00
		Std	69.47	63.71	63.79



BC Cancer Agency

CARE + RESEARCH

An agency of the Provincial Health Services Authority

Cancer Surveillance & Outcomes

# Customizing 2-D Table

- Edit headings and statistics keywords

```
proc tabulate data=sashelp.cars ;
class DRIVETRAIN ORIGIN TYPE ;
classlev DRIVETRAIN TYPE / s=[just=right];
keylabel COLPCTN='%' ;
var MSRP HORSEPOWER;
table
ALL (DRIVETRAIN ='Drive Type' TYPE) * (N={s=[just=right]}
COLPCTN={s=[just=right]})
(MSRP TYPE*HORSEPOWER) * (MEAN={s=[just=right]}
STD={s=[just=right]}),
(ORIGIN '=' ALL ='Total') / box='Car Stats';
run;
```

		Origin			Total
Car Stats		Asia	Europe	USA	Total
All	N	158	123	147	428
Drive Type					
All	N	34	36	22	92
	%	21.52	29.27	14.97	21.50
Front	N	99	37	90	226
	%	62.66	30.08	61.22	52.80
Rear	N	25	50	35	110
	%	15.82	40.65	23.81	25.70
Type					
Hybrid	N	3	.	.	3
	%	1.90	.	.	0.70
SUV	N	25	10	25	60
	%	15.82	8.13	17.01	14.02
Sedan	N	94	78	90	262
	%	59.49	63.41	61.22	61.21
Sports	N	17	23	9	49
	%	10.76	18.70	6.12	11.45
Truck	N	8	.	16	24
	%	5.06	.	10.88	5.61
Wagon	N	11	12	7	30
	%	6.96	9.76	4.76	7.01
MSRP	Mean	24741.32	48349.80	28377.44	32774.86
	Std	11321.07	25318.60	11711.98	19431.72
Type					
Hybrid	Horsepower	Mean	92.00	.	92.00
		Std	18.52	.	18.52
SUV	Horsepower	Mean	214.16	263.10	246.56
		Std	48.70	52.66	58.68
Sedan	Horsepower	Mean	181.98	236.53	191.99
		Std	57.29	71.35	46.50
Sports	Horsepower	Mean	225.35	316.74	312.00
		Std	57.60	96.21	91.09
Truck	Horsepower	Mean	190.25	.	242.13
		Std	51.76	.	60.48
Wagon	Horsepower	Mean	185.64	218.17	165.71
		Std	69.47	63.71	44.20



BC Cancer Agency

CARE + RESEARCH

An agency of the Provincial Health Services Authority

Cancer Surveillance & Outcomes

# Customizing 2-D Table

- Edit values (change formats, replace missing values)

```
proc tabulate data=sashelp.cars ;
class DRIVETRAIN ORIGIN TYPE ;
classlev DRIVETRAIN TYPE / s=[just=right];
keylabel COLPCTN='% ' ;
var MSRP HORSEPOWER;
table
ALL
(DRIVETRAIN='Drive Type' TYPE) * (N={s=[just=right]}
COLPCTN={s=[just=right]} *f=5.1)
(MSRP
TYPE*HORSEPOWER) * (MEAN={s=[just=right]} *f=dollar10.1
STD={s=[just=right]}),
(ORIGIN=' ' ALL='Total') / box='Car Stats' misstext='0';
run;
```

Car Stats			Asia	Europe	USA	Total
All		N	158	123	147	428
Drive Type						
All		N	34	36	22	92
		%	21.5	29.3	15.0	21.5
Front		N	99	37	90	226
		%	62.7	30.1	61.2	52.8
Rear		N	25	50	35	110
		%	15.8	40.7	23.8	25.7
Type						
Hybrid		N	3	0	0	3
		%	1.9	0	0	0.7
SUV		N	25	10	25	60
		%	15.8	8.1	17.0	14.0
Sedan		N	94	78	90	262
		%	59.5	63.4	61.2	61.2
Sports		N	17	23	9	49
		%	10.8	18.7	6.1	11.4
Truck		N	8	0	16	24
		%	5.1	0	10.9	5.6
Wagon		N	11	12	7	30
		%	7.0	9.8	4.8	7.0
MSRP		Mean	\$24,741.3	\$48,349.8	\$28,377.4	\$32,774.9
		Std	11321.07	25318.60	11711.98	19431.72
Type						
Hybrid	Horsepower	Mean	\$92.0	0	0	\$92.0
		Std	18.52	0	0	18.52
SUV	Horsepower	Mean	\$214.2	\$263.1	\$246.6	\$235.8
		Std	48.70	52.66	58.68	56.23
Sedan	Horsepower	Mean	\$182.0	\$236.5	\$192.0	\$201.7
		Std	57.29	71.35	46.50	62.80
Sports	Horsepower	Mean	\$225.4	\$316.7	\$312.0	\$284.2
		Std	57.60	96.21	91.09	92.79
Truck	Horsepower	Mean	\$190.3	0	\$242.1	\$224.8
		Std	51.76	0	60.48	61.85
Wagon	Horsepower	Mean	\$185.6	\$218.2	\$165.7	\$194.0
		Std	69.47	63.71	44.20	63.79



BC Cancer Agency

CARE + RESEARCH

An agency of the Provincial Health Services Authority

Cancer Surveillance & Outcomes

# Customizing 2-D Table

## ➤ Edit order of categories

```
proc format;
value $ drivefmt (notsorted)
'Front' = 'FRONT'
'Rear' = 'BACK'
'All' = 'ALL';
run;

proc tabulate data=sashelp.cars ;
class ORIGIN TYPE ;
class DRIVETRAIN/preloadfmt order=data;
classlev DRIVETRAIN TYPE / s=[just=right];
var MSRP HORSEPOWER;
table ALL (DRIVETRAIN='Drive Type' TYPE) * (N
COLPCTN='% '*f=5.1) (MSRP
TYPE*HORSEPOWER) * (MEAN*f=dollar10.1 STD),
(ORIGIN='' ALL='Total') /misstext='0'
box='Cars Stats';
format DRIVETRAIN $drivefmt.; run;
```

Car Stats			Asia	Europe	USA	Total
All	N		158	123	147	428
Drive Type						
FRONT	N		99	37	90	226
	%		62.7	30.1	61.2	52.8
BACK	N		25	50	35	110
	%		15.8	40.7	23.8	25.7
ALL	N		34	36	22	92
	%		21.5	29.3	15.0	21.5
Type						
Hybrid	N		3	0	0	3
	%		1.9	0	0	0.7
SUV	N		25	10	25	60
	%		15.8	8.1	17.0	14.0
Sedan	N		94	78	90	262
	%		59.5	63.4	61.2	61.2
Sports	N		17	23	9	49
	%		10.8	18.7	6.1	11.4
Truck	N		8	0	16	24
	%		5.1	0	10.9	5.6
Wagon	N		11	12	7	30
	%		7.0	9.8	4.8	7.0
MSRP		Mean	\$24,741.3	\$48,349.8	\$28,377.4	\$32,774.9
		Std	11321.07	25318.60	11711.98	19431.72
Type						
Hybrid	Horsepower	Mean	\$92.0	0	0	\$92.0
		Std	18.52	0	0	18.52
SUV	Horsepower	Mean	\$214.2	\$263.1	\$246.6	\$235.8
		Std	48.70	52.66	58.68	56.23
Sedan	Horsepower	Mean	\$182.0	\$236.5	\$192.0	\$201.7
		Std	57.29	71.35	46.50	62.80
Sports	Horsepower	Mean	\$225.4	\$316.7	\$312.0	\$284.2
		Std	57.60	96.21	91.09	92.79
Truck	Horsepower	Mean	\$190.3	0	\$242.1	\$224.8
		Std	51.76	0	60.48	61.85
Wagon	Horsepower	Mean	\$185.6	\$218.2	\$165.7	\$194.0
		Std	69.47	63.71	44.20	63.79




BC Cancer Agency

CARE + RESEARCH

An agency of the Provincial Health Services Authority

Cancer Surveillance & Outcomes

			Origin			All
			Asia	Europe	USA	
All	N		158	123	147	428
DriveTrain						
All	N		34	36	22	92
	ColPctN		21.52	29.27	14.97	21.50
Front	N		99	37	90	226
	ColPctN		62.66	30.08	61.22	52.80
Rear	N		25	50	35	110
	ColPctN		15.82	40.65	23.81	25.70
Type						
Hybrid	N		3	.	.	3
	ColPctN		1.90	.	.	0.70
SUV	N		25	10	25	60
	ColPctN		15.82	8.13	17.01	14.02
Sedan	N		94	78	90	262
	ColPctN		59.49	63.41	61.22	61.21
Sports	N		17	23	9	49
	ColPctN		10.76	18.70	6.12	11.45
Truck	N		8	.	16	24
	ColPctN		5.06	.	10.88	5.61
Wagon	N		11	12	7	30
	ColPctN		6.96	9.76	4.76	7.01
MSRP	Mean	24741.32	48349.80	28377.44	32774.86	
	Std	11321.07	25318.60	11711.98	19431.72	
Type						
Hybrid	Horsepower	Mean	92.00	.	.	92.00
		Std	18.52	.	.	18.52
SUV	Horsepower	Mean	214.16	263.10	246.56	235.82
		Std	48.70	52.66	58.68	56.23
Sedan	Horsepower	Mean	181.98	236.53	191.99	201.66
		Std	57.29	71.35	46.50	62.80
Sports	Horsepower	Mean	225.35	316.74	312.00	284.16
		Std	57.60	96.21	91.09	92.79
Truck	Horsepower	Mean	190.25	.	242.13	224.83
		Std	51.76	.	60.48	61.85
Wagon	Horsepower	Mean	185.64	218.17	165.71	194.00
		Std	69.47	63.71	44.20	63.79



Car Stats			Asia	Europe	USA	Total
All	N		158	123	147	428
Drive Type						
FRONT	N		99	37	90	226
	%		62.7	30.1	61.2	52.8
BACK	N		25	50	35	110
	%		15.8	40.7	23.8	25.7
ALL	N		34	36	22	92
	%		21.5	29.3	15.0	21.5
Type						
Hybrid	N		3	0	0	3
	%		1.9	0	0	0.7
SUV	N		25	10	25	60
	%		15.8	8.1	17.0	14.0
Sedan	N		94	78	90	262
	%		59.5	63.4	61.2	61.2
Sports	N		17	23	9	49
	%		10.8	18.7	6.1	11.4
Truck	N		8	0	16	24
	%		5.1	0	10.9	5.6
Wagon	N		11	12	7	30
	%		7.0	9.8	4.8	7.0
MSRP	Mean	\$24,741.3	\$48,349.8	\$28,377.4	\$32,774.9	
	Std	11321.07	25318.60	11711.98	19431.72	
Type						
Hybrid	Horsepower	Mean	\$92.0	0	0	\$92.0
		Std	18.52	0	0	18.52
SUV	Horsepower	Mean	\$214.2	\$263.1	\$246.6	\$235.8
		Std	48.70	52.66	58.68	56.23
Sedan	Horsepower	Mean	\$182.0	\$236.5	\$192.0	\$201.7
		Std	57.29	71.35	46.50	62.80
Sports	Horsepower	Mean	\$225.4	\$316.7	\$312.0	\$284.2
		Std	57.60	96.21	91.09	92.79
Truck	Horsepower	Mean	\$190.3	0	\$242.1	\$224.8
		Std	51.76	0	60.48	61.85
Wagon	Horsepower	Mean	\$185.6	\$218.2	\$165.7	\$194.0
		Std	69.47	63.71	44.20	63.79



BC Cancer Agency

CARE + RESEARCH

An agency of the Provincial Health Services Authority

Cancer Surveillance & Outcomes

# Exporting to Excel

---

```
ods tagsets.excelxp file = 'C:\My Documents\exampletable.xml'
  style=statistical
options(ABSOLUTE_COLUMN_WIDTH="10,10,6,6,6,6,6" EMBEDDED_TITLES='yes'
  SHEET_NAME='CARS');
title 'Descriptive Statistics for CARS dataset';
```

```
proc tabulate data=sashelp.cars ;
  class ORIGIN TYPE ;
  class DRIVETRAIN/preloadfmt order=data;
  classlev DRIVETRAIN TYPE / s=[just=right];
  keylabel COLPCTN='% ' ;
  var MSRP HORSEPOWER;
  table ALL (DRIVETRAIN='Drive Type' TYPE) * (N={s=[just=right]}
  COLPCTN={s=[just=right]}*f=5.1) (MSRP
  TYPE*HORSEPOWER) * (MEAN={s=[just=right]}*f=dollar10.1
  STD={s=[just=right]}), (ORIGIN=' ' ALL='Total')/box='Car Stats'
  misstext='0';
  format DRIVETRAIN $drivefmt.;
run;
```

```
ods tagsets.excelxp close;
```



**BC Cancer Agency**

CARE + RESEARCH

*An agency of the Provincial Health Services Authority*

**Cancer Surveillance & Outcomes**

exampletable.xml - Microsoft Excel

Home Insert Page Layout Formulas Data Review View

Clipboard Font Alignment Number Styles Cells Editing

Descriptive Statistics for CARS dataset

	A	B	C	D	E	F	G
1	Descriptive Statistics for CARS dataset						
2							
3	Car Stats			Asia	Europe	USA	Total
4	All	N		158	123	147	428
5	Drive Type						
6		FRONT	N	99	37	90	226
7			%	62.7	30.1	61.2	52.8
8		BACK	N	25	50	35	110
9			%	15.8	40.7	23.8	25.7
10		ALL	N	34	36	22	92
11			%	21.5	29.3	15	21.5
12	Type						
13		Hybrid	N	3	0	0	3
14			%	1.9	0	0	0.7
15		SUV	N	25	10	25	60
16			%	15.8	8.1	17	14
17		Sedan	N	94	78	90	262
18			%	59.5	63.4	61.2	61.2
19		Sports	N	17	23	9	49
20			%	10.8	18.7	6.1	11.4
21		Truck	N	8	0	16	24
22			%	5.1	0	10.9	5.6
23		Wagon	N	11	12	7	30
24			%	7	9.8	4.8	7
25	MSRP	Mean		\$24,741.3	\$48,349.8	\$28,377.4	\$32,774.9
26		Std		11321.07	25318.6	11711.98	19431.72
27	Type						
28		Hybrid Horsepower	Mean	\$92.0	0	0	\$92.0
29			Std	18.52	0	0	18.52
30		SUV Horsepower	Mean	\$214.2	\$263.1	\$246.6	\$235.8
31			Std	48.7	52.66	58.68	56.23
32		Sedan Horsepower	Mean	\$182.0	\$236.5	\$192.0	\$201.7
33			Std	57.29	71.35	46.5	62.8
34		Sports Horsepower	Mean	\$225.4	\$316.7	\$312.0	\$284.2
35			Std	57.6	96.21	91.09	92.79
36		Truck Horsepower	Mean	\$190.3	0	\$242.1	\$224.8
37			Std	51.76	0	60.48	61.85
38		Wagon Horsepower	Mean	\$185.6	\$218.2	\$165.7	\$194.0
39			Std	69.47	63.71	44.2	63.79

```
ods tagsets.excelxp file = 'C:\My Documents\exampletable.xml'
style=statistical
options(ABSOLUTE COLUMN WIDTH="10,10,6,6,6,6,6" EMBEDDED_TITLES='yes'
SHEET_NAME='CARS');
title 'Descriptive Statistics for CARS dataset';

proc tabulate data=sashelp.cars ;
class ORIGIN TYPE ;
class DRIVETRAIN/preloadfmt order=data;
classlev DRIVETRAIN TYPE / s=[just=right];
keylabel COLPCTN='% ' ;
var MSRP HORSEPOWER;
table ALL (DRIVETRAIN='Drive Type'
TYPE) * (N={s=[just=right]} COLPCTN={s=[just=right]}*f=5.1)
(MSRP TYPE*HORSEPOWER) * (MEAN={s=[just=right]}*f=dollar10.1
STD={s=[just=right]}), (ORIGIN=' ' ALL='Total')/box='Car
Stats' misstext='0';
format DRIVETRAIN $drivefmt.;

run;

ods tagsets.excelxp close;
```

Ready



BC Cancer Agency

CARE + RESEARCH

An agency of the Provincial Health Services Authority

Cancer Surveillance & Outcomes




```

proc format;
value $originfmt
'USA' = 'U:\My
Documents\BCCA\Presentations\VANSUG2013\usa.gif';
value numfmt
. = 'Orange'
0-200 = 'Blue'
201-1000 = 'Purple'
1001-high = 'Red'; run;

ods html file = 'U:\My Documents\exampletable.html';
proc tabulate data=sashelp.cars S=[foreground=numfmt.];
class TYPE / s=[background=pink foreground=blue];
class ORIGIN / s=[background=pink foreground=blue];
class DRIVETRAIN/preloadfmt order=data s=[background=pink
foreground=blue];
classlev DRIVETRAIN TYPE / s=[just=right
background=purple foreground=white];
classlev ORIGIN / S=[Vjust=T postimage=$originfmt.];
keylabel COLPCTN='%';
var MSRP HORSEPOWER;
table ALL (DRIVETRAIN='Drive Type'
TYPE)*(N={s=[just=right color=green]}
COLPCTN={s=[just=right color=red
background=orange]}*f=5.1) (MSRP
TYPE*HORSEPOWER)*(MEAN={s=[just=right]}*f=dollar10.1
STD={s=[just=right]})
, (ORIGIN=' ' ALL='Total')/box='Car Stats' misstext='0';
format DRIVETRAIN $drivefmt. ;
run;
ods html close;

```

		Asia	Europe	USA		Total
Car Stats						
All	N	158	123		147	428
Drive Type						
FRONT	N	99	37		90	226
	%	62.7	30.1		61.2	52.8
BACK	N	25	50		35	110
	%	15.8	40.7		23.8	25.7
ALL	N	34	36		22	92
	%	21.5	29.3		15.0	21.5
Type						
Hybrid	N	3	0		0	3
	%	1.9	0		0	0.7
SUV	N	25	10		25	60
	%	15.8	8.1		17.0	14.0
Sedan	N	94	78		90	262
	%	59.5	63.4		61.2	61.2
Sports	N	17	23		9	49
	%	10.8	18.7		6.1	11.4
Truck	N	6	0		16	24
	%	5.1	0		10.9	5.6
Wagon	N	11	12		7	30
	%	7.0	9.8		4.8	7.0
MSRP	Mean	\$24,741.3	\$48,349.8		\$28,377.4	\$32,774.9
	Std	11321.07	25318.60		11711.98	19431.72
Type						
Hybrid	Horsepower	Mean	\$92.0	0	0	\$92.0
		Std	18.52	0	0	18.52
SUV	Horsepower	Mean	\$214.2	\$283.1	\$248.6	\$235.8
		Std	48.70	52.66	58.68	56.23
Sedan	Horsepower	Mean	\$182.0	\$236.5	\$192.0	\$201.7
		Std	57.29	71.35	48.50	62.80
Sports	Horsepower	Mean	\$225.4	\$316.7	\$312.0	\$284.2
		Std	57.60	96.21	91.09	92.79
Truck	Horsepower	Mean	\$190.3	0	\$242.1	\$224.8
		Std	51.76	0	60.48	61.65
Wagon	Horsepower	Mean	\$185.6	\$218.2	\$165.7	\$194.0
		Std	69.47	63.71	44.20	63.79



**BC Cancer Agency**

CARE + RESEARCH

An agency of the Provincial Health Services Authority

**Cancer Surveillance & Outcomes**

# References

---

- <http://support.sas.com/resources/papers/proceedings09/039-2009.pdf>



**BC Cancer Agency**

CARE + RESEARCH

*An agency of the Provincial Health Services Authority*

**Cancer Surveillance & Outcomes**